// Copyright (C) 1999-2002, by Peter I. Corke
// Copyright (C) 2007  Interdepartmental Research Center "E. Piaggio", University of Pisa
//
// This file is part of RTSS, the Robotics Toolbox for Scilab/Scicos.
//
// RTSS is free software; you can redistribute it and/or modify
// it under the terms of the GNU General Public License as published by
// the Free Software Foundation; either version 2 of the License, or
// (at your option) any later version.
//
// RTSS is distributed in the hope that it will be useful,
// but WITHOUT ANY WARRANTY; without even the implied warranty of
// MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.  See the
// GNU General Public License for more details.
//
// You should have received a copy of the GNU General Public License
// along with RTSS; if not, write to the Free Software
// Foundation, Inc., 51 Franklin St, Fifth Floor, Boston, MA  02110-1301  USA



function [qdd] = rt_accel(robot, Q, qd, torque)
// File name:       rt_accel.sci
//
// Function:        rt_accel
//
// Description:     compute manipulator forward dynamics
//
// Annotations:     this code is a Scilab port of corresponding function in the
//                  Robotics Toolbox for MATLAB(R) written by Peter I. Corke.
//
// References:      Robotics Toolbox for MATLAB(R), robot7.1/accel.m
//
// Author:          Matteo Morelli, I.R.C. "E. Piaggio", University of Pisa
//
// Date:            April 2007
//
// $LastChangedDate: 2007-10-05 23:28:34 +0200(ven, 05 ott 2007) $

    [%nargout, %nargin] = argn(0);

    n = robot.n;
    if %nargin == 2 then

        q = Q(1:n);
        qd = Q(n+1:2*n);
        torque = Q(2*n+1:3*n);

    else

        q = Q;
        if length(q) == n then
            q = q(:);
            qd = qd(:);
        end

    end

    // compute current manipulator inertia
    M = rt_frne(robot, ones(n, 1)*q.', zeros(n, n), eye(n, n), [0; 0; 0]);

    // compute gravity and coriolis torque
    tau = rt_frne(robot, q.', qd.', zeros(1, n));

    qdd = inv(M) * (torque(:) - tau.');

endfunction
